Correction du TP20

Simulations d'expériences aléatoires

Exercice 1 1. Voici la fonction `deplacement` :

```
function res=deplacement(n)
    if n==1 then
        res=2
    elseif n==2 then
        if rand()<1/2 then
            res=1
        else
            res=3
        end
    else
        res=3
    end
endfunction
```

2. Voici la fonction `puce` :

```
function res=puce(n)
    res=1
    for k=1:n do
        res=deplacement(res)
    end
endfunction
```

3. Voici la fonction `arrivee` :

```
function res=arrivee()
    res=1
    n=deplacement(1)
    while n<>3 do
        res=res+1
        n=deplacement(n)
    end
endfunction
```

Exercice 2 1. L'univers des possibles associé à l'expérience aléatoire est $\Omega = \{1, 2, 3, 4, 5, 6\}$. Notons A_i l'événement "le dé donne le numéro i". Alors, d'après les hypothèses de l'énoncé, on a :

$$\begin{cases} P(A_1) + P(A_2) + P(A_3) + P(A_4) + P(A_5) + P(A_6) = 1 \\ P(A_1) = P(A_2) = P(A_3) = P(A_4) = P(A_5) \\ P(A_6) = 3P(A_1) \end{cases}$$

$$\Rightarrow \begin{cases} P(A_1) + P(A_1) + P(A_1) + P(A_1) + P(A_1) + 3P(A_1) = 1 \\ P(A_1) = P(A_2) = P(A_3) = P(A_4) = P(A_5) \\ P(A_6) = 3P(A_1) \end{cases}$$

$$\Rightarrow \begin{cases} P(A_1) = \frac{1}{8} \\ P(A_2) = P(A_3) = P(A_4) = P(A_5) = \frac{1}{8} \\ P(A_6) = \frac{3}{8} \end{cases}$$

Ainsi, $P(A_1) = P(A_2) = P(A_3) = P(A_4) = P(A_5) = \frac{1}{8}$ et $P(A_6) = \frac{3}{8}$.

2. Voici la fonction `de` qui simule le lancer du dé :

```
function res=de()
    if rand()<3/8 then
        res=6
    else
        res=floor(rand()*5+1)
    end
endfunction
```

3. (a) Voici la fonction `frequence1` qui calcule la fréquence d'apparition de chacun des numéros lors de n lancers du dé :

```
function F=frequence1(n)
    F=zeros(1,6)
    for k=1:n do
        x=de()
        F(x)=F(x)+1
    end
    F=F/n
endfunction
```

(b) On teste pour $n = 1000$:

```
-->frequence1(1000)
    0.126    0.131    0.106    0.122    0.129    0.386
```

On obtient le diagramme suivant :



On teste pour $n = 10000$:

```
-->frequence1(10000)
    0.1238    0.1278    0.1221    0.1232    0.1228    0.3803
```

On obtient le diagramme suivant :



On remarque que la fréquence d'apparition de chacun des numéros tend vers la probabilité d'apparition du numéro lorsque le nombre de lancers n est grand.

4. (a) Soit B l'événement : "le numéro obtenu est pair". On a $B = A_2 \cup A_4 \cup A_6$ et les événements A_2, A_4 et A_6 sont incompatibles. Donc :

$$P(B) = P(A_2 \cup A_4 \cup A_6) = P(A_2) + P(A_4) + P(A_6) = \frac{1}{8} + \frac{1}{8} + \frac{3}{8} = \frac{5}{8}.$$

(b) Voici la fonction `frequence2` pour obtenir la fréquence d'apparition d'un numéro pair lors de n lancers :

```
function res=frequence2(n)
    res=0
    for k=1:n do
        x=de()
        if floor(x/2)==x/2 then
            res=res+1
        end
    end
    res=res/n
endfunction
```

(c) On teste pour $n = 1000$ et pour $n = 10000$:

```
-->frequence2(1000)
   0.63
-->frequence2(10000)
   0.628
```

La fréquence d'apparition d'un numéro pair tend lorsque n est grand vers la probabilité d'obtenir un numéro pair, c'est-à-dire $\frac{5}{8} = 0.625$.

Exercice 3 1. (a) Voici la fonction `piece` :

```
function res=piece()
    res=0
    for k=1:10 do
        if rand()<1/2 then
            res=res+1
        end
    end
endfunction
```

(b) Voici la fonction `bouleblanche` :

```
function res=bouleblanche()
    k=piece()
    if rand()<k/10 then
        res=1
    else
        res=0
    end
endfunction
```

(c) Voici la fonction `frequence` :

```
function res=frequence(n)
    res=0
    for k=1:n do
        res=res+bouleblanche()
    end
    res=res/n
endfunction
```

On teste pour $n = 100$, pour $n = 1000$ et pour $n = 10000$:

```
-->frequence(100)
 ans  =
    0.55
-->frequence(1000)
 ans  =
    0.497
-->frequence(10000)
 ans  =
    0.5002
```

La fréquence d'apparition d'une boule blanche semble tendre vers $\frac{1}{2}$ lorsque n est grand.

2. (a) Il y a 2^{10} résultats possibles lors de 10 lancers d'une pièce (2 résultats possibles par lancer). Et il y a $\binom{10}{k}$ résultats favorables (le nombre de façons de choisir les k lancers où on fait pile parmi les 10 lancers effectués). Donc $P(A_k) = \dfrac{\binom{10}{k}}{2^{10}}$.

   (b) La famille $(A_k)_{0 \leq k \leq 10}$ forment un système complet d'évènements. Donc :
$$B = \bigcup_{k=0}^{10} (A_k \cap B).$$
   On obtient alors :
$$P(B) = P\left(\bigcup_{k=0}^{10} (A_k \cap B)\right) = \sum_{k=0}^{10} P(A_k \cap B) = \sum_{k=0}^{10} P(A_k) \times P_{A_k}(B),$$
   par incompatibilité des événements à la deuxième égalité et avec la formule des probabilités composées à la troisième.

   Comme $P(A_k) = \dfrac{\binom{10}{k}}{2^{10}}$ et $P_{A_k}(B) = \dfrac{k}{10}$, on a alors :
$$\begin{aligned} P(B) &= \sum_{k=0}^{10} \frac{\binom{10}{k}}{2^{10}} \frac{k}{10} = \frac{1}{2^{10} \times 10} \sum_{k=1}^{10} k \binom{10}{k} = \frac{1}{2^{10} \times 10} \sum_{k=1}^{10} 10 \binom{9}{k-1} = \frac{10}{2^{10} \times 10} \sum_{i=0}^{9} \binom{9}{i} \\ &= \frac{1}{2^{10}} \sum_{i=0}^{9} \binom{9}{i} 1^i 1^{9-i} = \frac{1}{2^{10}} \times 2^9 = \frac{1}{2}. \end{aligned}$$

   (c) Ce résultat est cohérent avec celui de la question 1.(c). La fréquence d'apparition d'une boule blanche tend vers la probabilité d'obtenir une boule blanche lorsque n est grand.

Exercice 4 1. Notons A_n l'événement "au moins deux étudiants ont leur anniversaire le même jour". Étudions plutôt la probabilité de l'événement $\overline{A_n}$ "les étudiants ont tous une date d'anniversaire différente". Il y a 365^n cas possibles (365 dates d'anniversaires possibles pour chaque étudiant) et $365 \times 364 \times \ldots \times (365 - n + 1)$ cas favorables (365 dates possibles pour le premier étudiant, 364 dates possibles pour le deuxième étudiant, ..., $365 - n + 1$ dates possibles pour le n-ième étudiant). Par équiprobabilité, on a donc :
$$\begin{aligned} P(\overline{A_n}) &= \frac{365 \times 364 \times \ldots \times (365 - n + 1)}{365^n} = \frac{365}{365} \times \frac{364}{365} \times \ldots \times \frac{365 - n + 1}{365} \\ &= 1 \times \left(1 - \frac{1}{365}\right) \times \ldots \times \left(1 - \frac{n-1}{365}\right) = \prod_{k=0}^{n-1} \left(1 - \frac{k}{365}\right). \end{aligned}$$

Donc :
$$p_n = P(A_n) = 1 - P(\overline{A_n}) = 1 - \prod_{k=0}^{n-1} \left(1 - \frac{k}{365}\right).$$

2. Voici la fonction `anniversaire` pour calculer p_n :

```
function p=anniversaire(n)
    p=1
    for k=0:n-1 do
        p=p*(1-k/365)
    end
    p=1-p
endfunction
```

3. Pour la classe d'ECE1 :

```
-->anniversaire(47)
    0.9547744
```

Il y a donc une probabilité de 0.95 que deux d'entre vous ait la même date d'anniversaire !

4. Calculons p_n pour différentes valeurs de n :

```
-->anniversaire(100)
    0.9999996927511
-->anniversaire(200)
    1.
-->anniversaire(300)
    1.
```

La suite $(p_n)_{n\geq 2}$ semble converger vers 1.

En fait, la suite $(p_n)_{n\geq 2}$ est stationnaire à partir du rang 366. En effet, si $n \geq 366$, on a :

$$\prod_{k=1}^{n-1}\left(1-\frac{k}{365}\right) = \left(1-\frac{365}{365}\right) \times \prod_{\substack{k=1 \\ k\neq 365}}^{n-1}\left(1-\frac{k}{365}\right) = 0 \times \prod_{\substack{k=1 \\ k\neq 365}}^{n-1}\left(1-\frac{k}{365}\right) = 0,$$

donc $\forall n \geq 366$, $p_n = 1$.

Ceci n'est pas surprenant ! Si on considère un groupe de plus de 366 personnes, au moins deux personnes ont la même date d'anniversaire (il y a 365 dates possibles pour plus de 366 personnes).

5. Voici la procédure pour déterminer le rang n à partir duquel p_n dépasse $\frac{1}{2}$:

```
n=2
while anniversaire(n)<1/2 do
    n=n+1
end
disp(n)
```

Alors :

```
-->exec('C:\TP20\exo4.sce', -1)
    23.
```

Ainsi, à partir de 23 élèves dans une classe, il y a plus d'une chance sur deux que deux élèves aient la même date d'anniversaire.